
Abstract

It has been shown that application of assembly implementation of Streaming
SIMD Extensions (SSE) shortens the time needed to apply filtration in two-
channel filter bank by tenfold, comparing to non-optimized version, written in
Microsoft Visual C++ 2010 Express, without assembler extensions. 

The implementation described in this paper can be applied to computation of
Discrete Wavelet Transform on general-purpose processors.. 

1 Introduction

Discrete Wavelet Transform (DWT) is applied to data compression, sys-
tem identification, signal approximation and interpolation, image processing 
and recognition as well as synthesis of digital watermarking [1-4].

Figure 1. Diagram of one stage of analysis and synthesis of Discrete Wavelet Trans-
form.

Because of such wide and profound applications, there is a lot of research 
on the improvements of Fast Computational Algorithms for the Discrete 
Wavelet Transform [5-10]. The construction of the algorithm is based on pa-
rallel or pyramidal repetition of basic analysis stage for forward transform and 
a basic synthesis stage for inverse transform. The two channel biorthogonal 
filter banks shown on Figure 1. [11] are a classic model of such a transform.

Blocks H, G, Q and R, are linear filters with finite impulse response H = 
h0,h1,...,hK-1, G = g0,g1,...,gK-1, Q = q0,q1,...,qK-1 i R = r0,r1,...,rK-1, where the
length of the filter K is an even natural number. Blocks ↓2 and ↑2 denote,
respectively, the operations of decimation in time of input sequence (down-
sampling) and upsampling by a factor of 2, i.e. inserting zeroes between each
sample of a input sequence.The results of analysis stage of (forward) DWT 
can be expressed as two convolutions with decimation [12]

where hK,k ,gK,k  for k = 0,1,...,K-1 are impulse responses of filters HK, GK, 
and N is the length of input sequence.  

If coefficients of impulse responses of filters HK and GK are written in re-
versed order:

h1K,k = hK,K-1-k, g1K,k = gK,K-1-k k = 0,1,...,K-1 

formulas (1) can be rewritten in the form (2) that is more convenient for 
implementation 

From (2) it is clear that the time needed for computation of DWT ex-
pressed as a convolution, depends on the effectiveness of floating point mul-
tiplications and additions. Exploiting Data Level Parallelism this can be en-
hanced by the usage of Streaming SIMD Extensions (SSE) available on con-
temporary general-purpose processors.

The paper describes construction of assembler implementations of DWT 
algorithms (2) that make use of SSE. The algorithms are given for a number 
of filter lengths K = 6,8,10 and 12 and the results are compared with the refer-
ence algorithms written in pure C++. 

The problem solved in the paper is important as the majority of personal 
computers in use, is equipped with processors that are compliant with SSE 
rather than newer AVX extensions, introduced in 2011 [13].

2 SSE in IA-32 architecture 

Beginning from the Pentium III processor the Streaming SIMD Extensions 
(SSE) were introduced to the IA-32 architecture. The SSE expands the SIMD 
execution model introduced with the Intel (Multimedia Extension) MMX 
technology by providing a new set of eight 128-bit registers xmm0, 
xmm1 ... xmm7 and the ability to perform (single-instruction, multiple-data) 
SIMD operations on four 32-bit packed single-precision floating-point values 
[13].The same operation can be performed at the same instruction cycle on 
four float elements stored in xmm register or in four array elements kept in 
memory. 

Because of this parallelism in data processing, application of SSE Exten-
sions can yield even fourfold performance gain comparing to a code that is 
non SSE aware. It is worth noting that data level parallelism reduces up to 
four times the number of instructions needed to write the algorithm.

3 Implementation of one stage of forward DWT computed as a 
convolution

Figure 2 shows the reference C++ implementation of DWT written accord-
ing to the formula (2).

// DWT in C++ 
for (int i=0;i<N;i+=2) 

 { 
float t1=h1[0]*x[i], t2=g1[0]*x[i]; 
for (int k=1;k<K;k++) 

  { 
t1+=h1[k]*x[i+k];  t2+=g1[k]*x[i+k]; 

  } 
y[i]=t1;  y[i+1]=t2; 

 } 

Figure 2. Algorithm of the one stage of forward DWT computed as a con-
volution in C++. 

The algorithm needs K floating-point multiplications and K-1 floating-
point additions to compute one output element. However because of the data 
level parallelism it is possible to significantly shorten the time of this compu-
tation by the application of SSE extensions.To maximize performance gain, 
the whole algorithm has been programmed in assembly lan-
guage.Furthermore, the inner loop that computes the sum of the product of
input values times coefficients of impulse response (in reversed order), has 
been unfolded and optimized for the selected filter lengths, to shorten the 
most computation intensive part. The outer loop that contains mainly instruc-
tions for reading samples and writing output coefficients has been left intact. 

Hence, further discussion in this section will concern major parts of the
two assembler implementations of forward DWT for N being divisible by 4, 
namely: version A, for filers of length K=6 and 8, version B, for K=10 and 12
as well as some elements of version C, for N being even and K=6. 

3.1 Version A. Implementation of DWT using assembler with SSE 
extensions 

The implementation of this version, for filter length K=8 is shown on Fig-
ure 3. For the sake of clarity and speed of computation it has been assumed
that the number of input samples N  is divisible by 4. It is not really a con-
straint as, in majority of DWT applications, the length of input sequence is 
power of 2 with the exponent greater than 1. However, this makes it possible 
to compute and keep four output coefficients in xmm register as well as store
them into the memory on every iteration of the loop. 

In the discussed implementation there are eight steps. The first step shown 
on part a) of Figure 3. loads four input samples x3, x2, x1, x0 into the register 
xmm0 and next four samples  x7, x6, x5, x4 into the register xmm1. It is illu-
strated by the comments to the code, where four parts of the relevant register 
are shown for every instruction. In part b) registers xmm4, xmm5, xmm6 i 
xmm7 are loaded with coefficients of impulse responses h1 and g1 in reversed
order. 

mov ecx,0   ;(i=0)ecx=0 
movaps xmm0,x[ecx] ; xmm0=x3|x2|x1|x0
movaps xmm1,x[ecx+16] ; xmm1=x7|x6|x5|x4
movaps buf1,xmm1 ; buf1=x7|x6|x5|x4

 Loading input data to the xmm registers

movaps xmm4,h1 ; xmm4=h13|h12|h11|h10
movaps xmm5,h1[16] ; xmm5=h17|h16|h15|h14
movaps xmm6,g1 ; xmm6=g13|g12|g11|g10

movaps xmm7,g1[16] ; xmm7=g17|g16|g15|g14

 Loading parameters h1 and g1 to the xmm registers 

iloop: 
movaps xmm2,xmm0 ; xmm2=xi+3|xi+2|xi+1|xi+0
mulps xmm2,xmm4 ; xmm2=xi+3*h13|xi+2*h12| 

; xi+1*h11|xi+0*h10
movaps xmm3,xmm1 ; xmm3=xi+7|xi+6|xi+5|xi+4
mulps xmm3,xmm5 ; xmm3=xi+7*h17|xi+6*h16| 

; xi+5*h15|xi+4*h14
addps xmm2,xmm3 ; xmm2=xi+7*h17+xi+3*h13| 

; xi+6*h16+xi+2*h12| 
;xi+5*h15+xi+1*h11|xi+4*h14+xi+0*h10

movaps xmm3,zeros ; xmm3=0.0|0.0|0.0|0.0 
haddps xmm2,xmm3 ; xmm2=0.0|0.0| 

;xi+7*h17+xi+3*h13+xi+6*h16+xi+2*h12| 
; xi+5*h15+xi+1*h11+xi+4*h14+xi+0*h10

haddps xmm2,xmm3 ; xmm2=0.0|0.0|0.0| 
; xi+7*h17+xi+6*h16+xi+5*h15+xi+4*h14+ 
; xi+3*h13+xi+2*h12+xi+1*h11+xi+0*h10

movaps buf5,xmm2 ; buf5=0.0|0.0|0.0|t1 

 Computation of coefficient t1 of DWT (at that moment it is yi+0)

movaps xmm2,xmm0 ; xmm2=xi+3|xi+2|xi+1|xi+0
mulps xmm2,xmm6 ; xmm2=xi+3*g13|xi+2*g12| 

; xi+1*g11|xi+0*g10
movaps xmm3,xmm1 ; xmm3=xi+7|xi+6|xi+5|xi+4
mulps xmm3,xmm7 ; xmm3=xi+7*g17|xi+6*g16| 

; xi+5*g15|xi+4*g14
addps xmm2,xmm3 ; xmm2=xi+7*g17+xi+3*g13| 

; xi+6*g16+xi+2*g12| 
; xi+5*g15+xi+1*g11| 
; xi+4*g14+xi+0*g10

movaps xmm3,zeros ; xmm3=0.0|0.0|0.0|0.0 
haddps xmm2,xmm3 ; xmm2=0.0|0.0| 

; xi+7*g17+xi+3*g13+ 
; xi+6*g16+xi+2*g12| 
; xi+5*g15+xi+1*g11+ 
; xi+4*g14+xi+0*g10

haddps xmm2,xmm3 ; xmm2=0.0|0.0|0.0| 
; xi+7*g17+xi+6*g16+ 

; xi+5*g15+xi+4*g14+xi+3*g13+ 
; xi+2*g12+xi+1*g11+xi+0*g10

shufps xmm2,xmm2,11110011b ; xmm2=0.0|0.0|t2|0.0 
addps xmm2,buf5 ; xmm2=0.0|0.0|t2|t1 
movaps buf5,xmm2 ; buf5=0.0|0.0|yi+1|yi+0

 Computation of coefficient t2 of DWT (at that moment it is yi+1)

movaps xmm2,x[ecx+32] ; xmm2=xi+11|xi+10|xi+9|xi+8
movaps buf0,xmm2 ; buf0=xi+11|xi+10|xi+9|xi+8
shufps xmm0,xmm1,01001110b ; xmm0=xi+5|xi+4|xi+3|xi+2
shufps xmm1,xmm2,01001110b ; xmm1=xi+9|xi+8|xi+7|xi+6

 Loading new input data to the xmm registers 

movaps xmm2,xmm0 ; xmm2=xi+5|xi+4|xi+3|xi+2
mulps xmm2,xmm4 ; xmm2=xi+5*h13|xi+4*h12| 

; xi+3*h11|xi+2*h10
movaps xmm3,xmm1 ; xmm3=xi+9|xi+8|xi+7|xi+6
mulps xmm3,xmm5 ; xmm3=xi+9*h17|xi+8*h16| 

; xi+7*h15|xi+6*h14
addps xmm2,xmm3 ; xmm2=xi+9*h17+xi+5*h13| 

; xi+8*h16+xi+4*h12| 
; xi+7*h15+xi+3*h11| 
; xi+6*h14+xi+2*h10

movaps xmm3,zeros ; xmm3=0.0|0.0|0.0|0.0 
haddps xmm2,xmm3 ; xmm2=0.0|0.0| 

; xi+9*h17+xi+5*h13+ 
; xi+8*h16+xi+4*h12| 
; xi+7*h15+xi+3*h11+ 
; xi+6*h14+xi+2*h10

haddps xmm3,xmm2 ; xmm3=0.0| 
; xi+9*h17+xi+8*h16+ 
; xi+7*h15+xi+6*h14+ 
; xi+5*h13+xi+4*h12+ 
; xi+3*h11+xi+2*h10| 
; 0.0|0.0

movaps buf6,xmm3 ; buf6=0.0|t1|0.0|0.0 

 Computation of coefficient t1 of DWT (at that moment it is yi+2)

movaps xmm2,xmm0 ; xmm2=xi+5|xi+4|xi+3|xi+2

mulps xmm2,xmm6 ; xmm2=xi+5*g13|xi+4*g12|xi+3*g11| 
; xi+2*g10

movaps xmm3,xmm1 ; xmm3=xi+9|xi+8|xi+7|xi+6
mulps xmm3,xmm7 ; xmm3=xi+9*g17|xi+8*g16|xi+7*g15| 

; xi+6*g14
addps xmm2,xmm3 ; xmm2=xi+9*g17+xi+5*g13| 

; xi+8*g16+xi+4*g12| 
; xi+7*g15+xi+3*g11| 
; xi+6*g14+xi+2*g10

movaps xmm3,zeros ; xmm3=0.0|0.0|0.0|0.0 
haddps xmm2,xmm3 ; xmm2=0.0|0.0| 

; xi+9*g17+xi+5*g13+ 
; xi+8*g16+xi+4*g12| 
; xi+7*g15+xi+3*g11+ 
; xi+6*g14+xi+2*g10

haddps xmm3,xmm2 ; xmm3=0.0|xi+9*g17+xi+8*g16+ 
; xi+7*g15+xi+6*g14+ 
; xi+5*g13+xi+4*g12+ 
; xi+3*g11+xi+2*g10|0.0|0.0

shufps xmm3,xmm3,10000000b ; xmm3=t2|0.0|0.0|0.0 
addps xmm3,buf6 ; xmm3=t2|t1|0.0|0.0 
addps xmm3,buf5 ; xmm3=yi+3|yi+2|yi+1|yi+0
movapsy[ecx],xmm3 ; y[ecx]=yi+3|yi+2|yi+1|yi+0

 Computation of coefficient t2 of DWT (at that moment it is yi+3), 
assembling yi+3, yi+2, yi+1, yi+0, in xmm register and storing its content 
into the memory

add ecx,16   ;(i=i+4) i.e. ecx=ecx+16 
movaps xmm0,buf1 ; xmm0=xi+7|xi+6|xi+5|xi+4
movaps xmm1,buf0 ; xmm1=xi+11|xi+10|xi+9|xi+8
movaps buf1,xmm1 ; buf1=xi+11|xi+10|xi+9|xi+8
cmpecx,NN ; Test the end of loop  

; condition(ecx = NN),  
;where NN=(N/4)*16 

jneiloop ; Jump to the label iloop 
; mentioned in step c)   
; if  ecx ≠ NN 

 updating xmm0 i xmm1 before the next iteration of the loop and exit from
the loop.

Figure 3. Steps of computation of the coefficients yi+3, yi+2, yi+1, yi+0 for i = 
0,4,8,...,N-4 in Version A of the implementation of DWT using assembler
with SSE extensions 

After the initialization steps a) - b) the algorithm enters the loop shown in 
steps c) - h). The step c) shows how the output coefficient yi+0  is being com-
puted and put into the least significant part of the xmm register. Two SSE
multiplications and three SSE additions are performed in this phase that is 
equivalent to eight floating point multiplications and seven additions. The 
results are saved to appropriate parts of the xmm register. 

The next phase of the loop, namely step d) is devoted to computation of
output coefficient yi+1 and  saving it in the subsequent, more significant part of 
the xmm register. After this step, the register contains coefficients yi+1, yi+0in 
its lower part and floating zeros in the upper part. 

This step is almost identical to the preceding one, with the exception of re-
placing impulse response h1with g1.

In step e) registers xmm0 and xmm1 are loaded with input samples shifted 
by two positions, in relation to their previous content. Namely,xmm0 contains 
samples x5, x4 ,x3 ,x2, and xmm1samples x9, x8, x7, x6. These data will be used to 
compute yi+3, yi+2. 

Step f) show the details of computation of  coefficient yi+2 and points out 
that its value is stored in the xmm register, next to already computed yi+1, yi+0.
Again, this phase is very similar to step c), the only difference is the position 
in xmm register where the value of yi+2 is being saved. 

Similarly, the step g) concerns computation of the forth coefficient yi+3. It
is stored in the most significant part of the register xmm. The final quadruplet
yi+3, yi+2, yi+1, yi+0 is saved from the xmm register into the memory. 

The last phase of loop shown as step h) increments loop counter in ecx reg-
ister by 16 (i.e. the size of xmm register in bytes) This value will be used to 
address input samples x and output coefficients y. 

In the following lines of code, registers xmm0 and xmm1 are being pre-
pared for computation of output coefficients yi+3, yi+2, yi+1, yi+0 in the next 
iteration of the loop (with i incremented by 4). 

The loop concludes with a test condition ecx ≠ NN, where NN=(N/4)*16.
If this condition is met the code jumps to the label iloop discussed in

step c), i.e. the beginning of the loop. 
The code for case K=8 can be also used for K=6, provided the two most 

significant coefficients of reversed order impulse responses are set to zero, i.e. 
h17=h16=g17=g16=0. However, coefficients h15� h10 and  g15� g10 need to 
be initialized with corrected values, appropriate for K=6. 

3.2 Version B. Implementation of DWT using assembler with SSE
extensions 

The implementation for filter length K=12 is similar to Version A. Again it
has been assumed that the number of input samples N  is divisible by 4, which
makes it possible to process, in one iteration of the loop, four output coeffi-
cient in the register xmm. 

The implementation can be logically divided into 8 steps. At first, the reg-
isters xmm0 are loaded with values x3, x2, x1, x0, xmm1 with x7, x6, x5, x4 and 
xmm2 with x11, x10, x9, x8. The reversed coefficients of impulse responses h17, 
h16, h15, h14and h13, h12, h11, h10 are sent to xmm5 and xmm4, while g17, g16, 
g15, g14and g13, g12, g11, g10 are sent to xmm7 and xmm6. However, because
of the limited number of xmm registers the most significant parts of h111, h110, 
h19, h18 and g111, g110, g19, g18 will be fetched from memory. 

Following the above initialization code there are six steps in the loop, as 
they were in version A.The first step shows how the output coefficient yi+0  is
being computed and put into the least significant part of the xmm register.
Three SSE multiplications and four SSE additions are performed in this phase 
which is equivalent to twelve floating point multiplications and eleven addi-
tions. The results are saved to appropriate parts of the xmm register.

The next phase of the loop is devoted to computation of output coefficient
yi+1 and  saving it into the subsequent, more significant part of the xmm regis-
ter. After this step, the register contains coefficients yi+1, yi+0in its lower part 
and floating zeros in the upper part. Thatphase of the algorithm is almost iden-
tical to the preceding one, with the exception of replacing impulse response 
h1with g1.  

In the next step registers xmm0, xmm1 and xmm2 are loaded with input 
samples shifted by two positions, in relation to their previous content. Name-
ly,xmm0 contains samples x5, x4 ,x3 ,x2,, xmm1samples x9, x8, x7, x6, andxmm2 
samples x13, x12, x11, x10.  These data will be used to compute yi+3, yi+2. 

In the subsequent step, coefficient yi+2 is being computed and stored in the 
xmm register, next to the already computed yi+1, yi+0. Again, this phase is very 
similar to the computation of yi+0, the only difference is the position in xmm
register where the value of yi+2 is being saved. 

Similarly, the following step, concerns computation of the last coefficient 
yi+3. It is stored in the most significant part of the register xmm. The final qua-
druplet yi+3, yi+2, yi+1, yi+0 is transferred from the xmm register into the memo-
ry. 

The last phase of loop increments loop counter in ecx register by 16 (i.e. 
the size of xmm register in bytes). This value will be used to address input
samples x and output coefficients y. 

In the following lines of code, registers xmm are being prepared for com-
putation of output coefficients yi+3, yi+2, yi+1, yi+0 in the next iteration of the 

loop (with i incremented by 4). The loop concludes with a test condition ecx ≠
NN, where NN=(N/4)*16.If this condition is met the code jumps to the label 
iloop, i.e. computation of yi+0.  

The code that computes DWT for K=12 can be also used for K=10, pro-
vided the most significant coefficients of reversed impulse responses are set to
zero, i.e. h111,=h110,=g111,=g110=0. However, coefficients h19� h10 and 
g19� g10 need to be initialized with corrected values, appropriate for K=10.

3.3 Version C. Implementation of DWT using assembler with SSE 
extensions 

This version of DWT implementation assumes that K=6, but the number of 
input samples Nis even. In that case, on every odd iteration of the loop, the 
computations are performed in the same way as in version A (see Figure 3.
Step c), and the pair of output coefficients yi+1, yi+0 is saved on the least signif-
icant positions od 128-bit long buffer buf. On every even iteration, a following 
pair of output coefficients is being computed and saved in memory, together
with a preceding pair, as a quadruplet of properly ordered coefficients. 

If N is not divisible by 4, the last save operation concerns only the last pair 
of yi+1, yi+0.The code for computation of output coefficients is identical in 
versions A and C of the algorithm, so is the time of computation for K=6. 

4 Test environment 

All DWT implementations presented in the paper were written as C++ in-
line assembly (with SSE extensions) and compiled with Microsoft Visual C++ 
2010 Express Version 10.0.40219.1 SP1Rel. The compiled code was executed 
on MS Windows 7 Home Premium PC with Intel® Core� i5 CPU 650 
3.20GHz and 4GB of RAM on board. Further, to neglect impact of concurrent
operations of the processor on the computation time, all tests were run pN
times, and the minimum time of execution, obtained with 64-bit clock cycle 
counter (measuring the number of clock cycles of the very code responsible 
for the computation of DWT), has been taken as a actual result of the mea-
surement.

5 Experimental results

In order to compare effectiveness of the proposed implementation of DWT 
using assembler with SSE extensions, it was compared against the reference 
program written in pure C++, for the selected lengths of filter K, and a few 

lengths of a input sequence N being divisible by 4. The resulting measure-
ments expressed in cycles are gathered in Table 1. 

Table 1. Results of measurement for pN = 100 000 000

K Implementation N=64 N=256 N=1024 N=4096 

6 
C++ 3 465 13 983 55 902 223 611
Assembler with SSE
(version A) 

459 1 845 7 395 29 490

8 
C++ 4 479 18 246 72 927 291 693
Assembler with SSE
(version A) 

459 1 845 7 389 29 862

10
C++ 5 577 22 374 89 193 357 006
Assembler with SSE
(version B) 

594 2 403 9 621 38 880

12
C++ 6 579 26 628 106 965 425 748
Assembler with SSE
(version B) 

597 2 424 9 657 38 616

As can be seen from the table above, for filters of length K=8 and all tested
values of N, implementation of DWT in assembler, with SSE extensions is 
performed almost 10 times faster than pure C++ version. For K=12 the opti-
mized code is almost 11 times faster. This speedup may be attributed to ma-
nually optimized assembler implementation with parallel processing of data 
using SSE extensions. As it was mentioned in the information about SSE in
IA-32 architecture, this may shorten the time of computation up to four times. 

Further reduction of execution time, results from unfolding the inner loop 
which is the most computationally intensive. The outer loop contains mainly
instructions for reading samples x and writing output coefficients y. 

Because the implementation for K=6 and 8 uses the same version (A) of 
the algorithm, execution time is almost identical in both cases. The same
holds true for version B and K=10 and 12.

Regardless of the version of implementation and the value of K, the
amount of time needed to compute DWT is proportional to the length of the 
input sequence (and number of iterations). It is a direct conclusion from the
formula (2).

Eventually the version C, for even N, has been examined. The results are
shown in Table 2. 

Table 2. Results of measurement for pN = 100 000 000

Implementation N=64 N=256 N=1024 N=4096 
Assembler with SSE (version C) 465 1839 7389 29436 

The execution times for version C are virtually identical to version A. As a 
matter of fact, it is an expected result as both implementations share the same
code to compute output pairs of coefficients. Moreover, although the con-
struction and analysis of version C is more complex than version A, the speed
of version C remains the same. Therefore, it is sufficient to use version A for 
K=6 and 8, and version B for K=10 and 12 and exclude special implementa-
tions for N, that are even but not divisible by four.

6 Conclusions 

The paper discusses a number of implementations of Discrete Wavelet 
Transform written as a formula (2). The experimental results show that ma-
nually optimized C++, with unfolded inner loop and inline assembly code
with SSE extensions, is about 10 times more robust than reference program 
written in pure C++. What is more, the obtained speedup looks favorably, 
comparing to the results shown in [14] where the SSE enabled code was per-
formed 6x faster than naïve, C++  implementation of the convolution algo-
rithm.  

Although it is possible to achieve even further speedup with the applica-
tion of the thread level parallelism of contemporary multi-core processors, the 
necessary algorithms are considerably more complicated. Hence, the proposed
solution that use only Data Level Parallelism with SSE extensions is an attrac-
tive alternative, available even on a simple one core processors.

Due to the lower complexity of versions A and B, they are recommended 
as effective templates for computation of DWT with application of SSE.  

